Práctica 5

5.1 Curvas en coordenadas paramétricas. Representación gráfica.

Para representar gráficamente una curva plana dada por sus ecuaciones paramétricas escribiremos

```
-->     wxplot2d([parametric,cos(t),sin(t),[t,-%pi,%pi]])
```

se pueden combinar con curvas explícitas

```
-->     wxplot2d ([x^3+2,[parametric,cos(t),sin(t),[t,-%pi, %pi]]],[x,-1,2])$
```

Para curvas paramétricas en el espacio, vamos a cargar primero el paquete `draw`

```
-->     load(draw)$

-->     draw3d(color = royalblue,
               key   = "Curva 1",
               parametric(cos(2*u)^2,sin(3*u),u-2,u,0,2),
             color = red,
             key = "Curva 2",
             line_width = 3,
             parametric(t^2,sin(t),2+t,t,0,2),
        title = "Curvas")$
```

donde se ve cómo las opciones para cada curva están escritas delante de sus ecuaciones. También se pueden combinar curvas y superficies

```
-->     f(x,y):=x*y*%e^(-x^2-y^2)$
        draw3d(color=green,explicit( f(x,y),x,-2,2,y,-2,2),
                    color=red,
                    line_width = 2,
                    parametric(t,t,f(t,t),t,-2,2));
```

Este mismo paquete, se puede utilizar para curvas planas

```
-->     draw2d(explicit(exp(x),x,-1,3),
           color = red,
        parametric(2*cos(t),t^2,t,0,2*%pi))$
```

5.1.1 Estudio de una curva plana para su representación gráfica

Se introducen las ecuaciones de la curva,

```
-->     x(t):=-(t+2)/t$ y(t):=(t+1)*(t-2)/t^2$
```

Su representación se obtendría mediante la sentencia

```
-->     wxplot2d([parametric,x(t),y(t),[t,-5,6] ]);
```

Para el estudio, calculamos las derivadas

```
-->     xp:factor(diff(x(t),t,1))$ yp:factor(diff(y(t),t,1))$
```

y buscamos los valores de t que hacen 0 o ∞ una de las cuatro funciones x, x', y, y'. Para ello, buscamos los valores reales que anulan alguno de los numeradores o denominadores, por lo que primero resolveremos ecuaciones del tipo

```
   solve(x(t),t);solve(denom(x(t)),t);
```

Para que todos esos valores estén en una misma lista escribimos

```
-->     flatten( [solve(x(t),t),solve(denom (x(t)),t),
        solve(y(t),t),solve(denom (y(t)),t),
        solve(xp,t),solve(denom(xp),t),
        solve(yp,t),solve(denom(yp),t)]);
```

La orden

```
-->     v: unique(%);
```

suprime las repeticiones. Para poder utilizar la lista anterior adecuadamente, debemos eliminar, de cada elemento, la expresión `t =` . Para ello, formamos una lista con los términos de la derecha de cada una de esas igualdades

```
-->     valores:makelist(rhs(v[i]),i,1,length(v));
```

En la lista pueden aparecer números complejos, dependiendo de las ecuaciones, en ese caso los suprimimos con

```
for k:length(valores) thru 1 step -1 do ( if(imagpart(valores)[k])#0
then ( valores:delete(valores[k],valores)));
```

Para finalizar, calculamos las columnas de la tabla correspondientes a esos valores de t

```
-->     for k:1 thru length(valores) do (print(valores[k],
        limit([x(t),y(t),xp,yp,yp/xp],t,valores[k])));
```

y, para las columnas primera y última de la tabla, escribimos también

```
-->     limit([x(t),y(t),xp,yp,yp/xp],t,-infinity);
        limit([x(t),y(t),xp,yp,yp/xp],t,+infinity);
```

Ejercicio 1. Completar el cuadro siguiente.

t	$-\infty$		-4		-2		-1		0		2		$+\infty$
$x(t)$													
$y(t)$													
$x'(t)$													
$y'(t)$													
dy/dx													

5.2 Triedro intrínseco

Para algunos cálculos con vectores es necesario cargar el paquete

```
-->     load ("vect");
```

Si existe el triedro intrínseco de curva $\gamma \equiv r = \overrightarrow{r}(t)$ en un punto, sabemos que un vector direccional de la recta tangente (y característico del plano normal) es

$$\overrightarrow{r'}(t).$$

Así, si en $P(a) \in \gamma$ existe recta tangente, sus ecuaciones paramétricas son

$$\left.\begin{array}{l} x = x(a) + \lambda x'(a) \\ y = y(a) + \lambda y'(a) \\ z = z(a) + \lambda z'(a) \end{array}\right\} \lambda \in I\!R.$$

Un vector (característico del plano osculador y) direccional de la recta binormal es:

$$\overrightarrow{r'}(t) \wedge \overrightarrow{r''}(t),$$

y un (característico del plano rectificante y) direccional de la normal principal:

$$\overrightarrow{r'}(t) \wedge (\overrightarrow{r'}(t) \wedge \overrightarrow{r''}(t)).$$

Escribimos las ecuaciones de γ y vamos a hacer los cálculos en el punto asociado al parámetro $a = 0$

```
-->     x(t):=cos(t)$          y(t):=sin(t)$        z(t):=2*t*cos(t)$
        r(t):=[x(t),y(t),z(t)]$
        a:0$
```

Calculamos el vector derivada primera de $\overrightarrow{r}(t)$

```
-->     dp:diff(r(t),t)$
```

y así, la recta tangente es la que pasa por $\overrightarrow{r}(a)$ y tiene un direccional $\overrightarrow{vt} = \overrightarrow{r'}(a)$

```
-->     vt:dp,t=a;
```

con lo que la gráfica de la curva con la tangente en $P(0)$ será

```
-->     draw3d(color = royalblue,
               key   = "Curva",
               parametric(r(t)[1],r(t)[2],r(t)[3],t,-1,1),
        color = red, key   = "tg", line_width = 3,
        parametric(r(a)[1]+u*vt[1],
        r(a)[2]+u*vt[2],r(a)[3]+u*vt[3],u,-1,1))$
```

Para obtener un vector direccional de la binormal en ese punto, calculamos el vector derivada segunda de $\overrightarrow{r}(t)$

```
-->     ds:diff(r(t),t,2)$
```

y un vector direccional en $t = a$ será

```
-->     vb:dp ~ ds; express(vb);
        vb:%,t=a;
```

y un vector direccional de la normal

```
-->     vn:vb ~ vt; vn:express(vn);
```

Ejercicio 2. Representar la curva junto con las tres aristas del triedro.

Con los vectores característicos de las caras del triedro ya calculados anteriormente

$$vt = [0, 1, 2], \quad vb = [0, -2, 1] \ \text{ y } \ vn = [-5, 0, 0],$$

para obtener sus ecuaciones escribimos:

```
-->     vecpl:[X,Y,Z]-r(a);
```

para el plano normal

```
-->     plno(X,Y,Z):=vt.vecpl$
```

para el osculador:

```
-->     plos(X,Y,Z):=vb.vecpl$
```

y para el rectificante:

```
-->     plre(X,Y,Z):=vn.vecpl$
```

La representación gráfica de la curva, junto con los planos normal y osculador se obtiene con

```
-->     draw3d(
        color = royalblue,
           key    = "Curva",line_width = 3,
           parametric(r(t)[1],r(t)[2],r(t)[3],t,-1,1),
        color = red, key    = "Plano normal", line_width = 1,
          implicit(plno(X,Y,Z),X,-1,1,Y,-1,1,Z,-1,1),
        color = green, key    = "Plano osculador",
          implicit(plos(X,Y,Z),X,-1,1,Y,-1,1,Z,-1,1)
          )$
```

sin embargo, la ecuación del plano rectificante queda $-5(X-1)=0$, es decir $X=1$, por lo que los puntos de este plano son de la forma $(1,Y,Z)$ y no admite la ecuación del mismo modo que los anteriores pues la primera coordenada no varía. Así, para representar también este plano vertical, añadiremos al bloque anterior

```
-->     draw3d(color = royalblue,
               ...
               ...    ,
        color = violet,key    = "Plano rectificante",
        parametric_surface(1,Y,Z,Y,-1,1,Z,-1,1)
          )$
```

Ejercicio 3. Representar la curva anterior junto con las aristas del triedro para $t=1$.

5.3 Curvatura y Torsión

Para la obtención de $\kappa = \dfrac{|\overrightarrow{r'}(t)\wedge\overrightarrow{r''}(t)|}{|\overrightarrow{r'}(t)|^3}$ y $\tau = \dfrac{[\overrightarrow{r'}(t),\overrightarrow{r''}(t),\overrightarrow{r'''}(t)]}{|\overrightarrow{r'}(t)\wedge\overrightarrow{r''}(t)|^2}$ necesitamos el módulo de un vector. Para ello, introducimos la función

```
-->     modulo(v):= if listp(v) then sqrt(apply("+",v^2))
                                else error(v, "no es un vector")$
```

Ejercicio 4. Calcular la curvatura y la torsión, para $t=\pi/2$, de la hélice circular de ecuaciones

$$x(t)=4\cos t,\ y(t)=4\,\mathrm{sen}\,t,\ z(t)=6t.$$

5.3.1 Curvatura de una curva plana

Si γ está en el plano $z=0$, $\kappa = \dfrac{\begin{vmatrix} x'(t) & y'(t) \\ x''(t) & y''(t) \end{vmatrix}}{(x'(t)^2+y'(t)^2)^{3/2}}$

Ejercicio 5. Hallar la curvatura de la elipse de semiejes 3 y 2 para $t = \pi/4$.

$$x(t) = 3\cos(t), y(t) = 2\,\mathrm{sen}\,(t)$$

En el caso particular en el que la ecuación de γ es de la forma $y = y(x)$, es decir: $x(t) = t, y = y(t), z(t) = 0$, queda $\kappa = \dfrac{y''(x)}{(1 + y'(x)^2)^{3/2}}$.

Ejercicio 6. Hallar la curvatura de la parábola $y = x^2$ en el punto de abscisa $x = 1/2$.

5.4 Evoluta

La ecuación de la evoluta de una curva alabeada γ es

$$\overrightarrow{x} = \overrightarrow{r}(t) - \frac{|\overrightarrow{r'}(t)|^2}{|\overrightarrow{r'}(t) \wedge \overrightarrow{r''}(t)|^2}\left(\overrightarrow{r'}(t) \wedge (\overrightarrow{r'}(t) \wedge \overrightarrow{r''}(t))\right).$$

Si la curva está en el plano $z = 0$, la ecuación resulta

$$X = \left(x(t) - \frac{x'(t)^2 + y'(t)^2}{\begin{vmatrix} x'(t) & y'(t) \\ x''(t) & y''(t) \end{vmatrix}} y'(t) \right) \overrightarrow{\imath} + \left(y(t) + \frac{x'(t)^2 + y'(t)^2}{\begin{vmatrix} x'(t) & y'(t) \\ x''(t) & y''(t) \end{vmatrix}} x'(t) \right) \overrightarrow{\jmath}$$

si además la ecuación de γ es de la forma $y = y(x)$ las ecuaciones paramétricas de la evoluta quedan

$$\left.\begin{array}{l} X = x - \dfrac{1 + y'(x)^2}{y''(x)} y'(x) \\ Y = y(x) + \dfrac{1 + y'(x)^2}{y''(x)} \\ Z = 0 \end{array}\right\}$$

Ejercicio 7. Ecuación de la evoluta de la elipse del ejercicio 5. Representar ambas curvas.